/* -*- Mode: C++; tab-width: 8; indent-tabs-mode: nil; c-basic-offset: 2 -*- */
/* This Source Code Form is subject to the terms of the Mozilla Public
 * License, v. 2.0. If a copy of the MPL was not distributed with this
 * file, You can obtain one at http://mozilla.org/MPL/2.0/. */

include protocol PNecko;
include protocol PBrowser;
include URIParams;
include PBackgroundSharedTypes;
include PBrowserOrId;

using class IPC::SerializedLoadContext from "SerializedLoadContext.h";

namespace mozilla {
namespace net {

//-------------------------------------------------------------------
protocol PWyciwygChannel
{
  manager PNecko;

parent:
  async __delete__();

  async Init(URIParams     uri,
             PrincipalInfo requestingPrincipalInfo,
             PrincipalInfo triggeringPrincipalInfo,
             PrincipalInfo principalToInheritInfo,
             uint32_t      securityFlags,
             uint32_t      contentPolicyType);
  async AsyncOpen(URIParams             originalURI,
                  uint32_t              loadFlags,
                  SerializedLoadContext loadContext,
                  PBrowserOrId browser);
  async AppData(SerializedLoadContext loadContext, PBrowserOrId browser);

  // methods corresponding to those of nsIWyciwygChannel
  async WriteToCacheEntry(nsString data);
  async CloseCacheEntry(nsresult reason);
  async SetCharsetAndSource(int32_t source, nsCString charset);
  async SetSecurityInfo(nsCString securityInfo);
  async Cancel(nsresult status);

child:
  async OnStartRequest(nsresult  statusCode,
                       int64_t   contentLength,
                       int32_t   source,
                       nsCString charset,
                       nsCString securityInfo);

  async OnDataAvailable(nsCString data,
                        uint64_t  offset);

  async OnStopRequest(nsresult statusCode);

  async CancelEarly(nsresult statusCode);
};


} // namespace net
} // namespace mozilla
